**Assessing the Assumption of Equal Variances - A Revisit**

Up to this point we have considered using either Levene's test for assessing the assumption of equal variances or a plot of the residuals versus the treatment levels. It is now time to introduce SAS Proc Mixed, which performs the computations for ANOVA, but also allows one to assess the structure of the variance and covariance among factor levels. The syntax structure of Proc Mixed is similar to that of Proc GLM, but there are differences.

For example, consider the following problem which was previously discussed (page 17, 18):

Seventy-eight (78) male workers were assigned at random to six different groups so that 13 were in each group. After training in a specific task, the pulse rate was measured for 20 seconds. Unfortunately, some (10) individuals withdrew from the experiment before their training was complete. The data from this experiment, along with some summary statistics are reproduced in the following table:

| Group | | | | | |
|---|---|---|---|---|---|
| 1 | 2 | 3 | 4 | 5 | 6 |
| 27 | 29 | 34 | 34 | 28 | 28 |
| 31 | 28 | 36 | 34 | 28 | 26 |
| 26 | 37 | 34 | 43 | 26 | 29 |
| 32 | 24 | 41 | 44 | 35 | 25 |
| 39 | 35 | 30 | 40 | 31 | 35 |
| 37 | 40 | 44 | 47 | 30 | 34 |
| 38 | 40 | 44 | 34 | 34 | 37 |
| 39 | 31 | 32 | 31 | 34 | 28 |
| 30 | 30 | 32 | 45 | 26 | 21 |
| 28 | 25 | 31 | 28 | 20 | 28 |
| 27 | 29 | | | 41 | 26 |
| 27 | 25 | | | 21 | |
| 34 | | | | | |

The analysis outlined on page 18 indicated that the assumption of equal variance was satisfied by Levene's test (P = 0.6688). In addition, the plot of the residuals versus the treatment groups showed no particular deviation either.

Letting "Group" represent the treatment factor (levels 1 through 6) and Replicate represent the individuals from each group (1 through 13 for group 1, 1 through 12 for group 2, etc.) the SAS Proc GLM Code for analyzing these data would be:

```
Proc GLM data = Task;
 Class Group Replicate;
  Model Response = Group;
run;
```

This analysis would produce the usual ANOVA table for assessing the global null hypothesis of no difference among the Group means. To analyze these data using Proc Mixed the following code might be used:

```
Proc Mixed data = Task;
 Class Group Replicate;
  Model Response = Treatment;
  Repeated / type = un(1) subject = Replicate r;
run;
```

Aside from the change in procedure used, the only difference is the use of the "Repeated" statement. This statement invokes a method for allowing the "subject" variable to have a variance (and correlation) structure which differs from the usual assumption of equal variances and no correlation between observations.

The "Repeated" statement shown above ***"Repeated / type = un(1) subject = Replicate r;***" allows for a covariance structure of type "***un(1)***". The "un" stands for unstructured covariance matrix and the 1 in parentheses indicates that only the variances are allowed to differ, but that there is no covariance or correlation between treatments. SAS Proc Mixed allows for many different structure for the covariance matrix. These include auto-regressive, spatial, compound symmetric and unstructured types. In stead of listing the possible structures it would be best of you reviewed the Proc Mixed syntax in the online SAS help menu. The"***r***" at the end of the statement tells SAS to print the covariance structure for review.

The analysis of the Pulse data using Proc Mixed with unequal variances is as follows:

```
options pageno = 1;
title "Analysis of Pulse Rate for 6 Treatment Groups";
data task;
input Pulse Group Replicate @@;
cards;
27 1  1   29 2  1   34 3  1    34 4  1    28 5  1    28  6  1
31 1  2   28 2  2   36 3  2    34 4  2    28 5  2    26  6  2
26 1  3   37 2  3   34 3  3    43 4  3    26 5  3    29  6  3
32 1  4   24 2  4   41 3  4    44 4  4    35 5  4    25  6  4
39 1  5   35 2  5   30 3  5    40 4  5    31 5  5    35  6  5
37 1  6   40 2  6   44 3  6    47 4  6    30 5  6    34  6  6
38 1  7   40 2  7   44 3  7    34 4  7    34 5  7    37  6  7
39 1  8   31 2  8   32 3  8    31 4  8    34 5  8    28  6  8
30 1  9   30 2  9   32 3  9    45 4  9    26 5  9    21  6  9
28 1 10   25 2 10   31 3 10    28 4 10    20 5 10    28  6 10
27 1 11     29 2 11                        41 5 11    26  6 11
27 1 12     25 2 12                        21 5 12
34 1 13
;
proc print data = task;
run;
title2 "Analysis of Variance for Using Proc Mixed";
Proc Mixed data = task;
  Class Group Replicate;
  Model Pulse = Group;
  Repeated / type = un(1) subject = Replicate r;
run;
```

The resulting analysis is, in part, as follows:

```
        Analysis of Pulse Rate for 6 Treatment Groups                    1
          Analysis of Variance for Using Proc Mixed
                                              10:44 Friday, March 21, 2008

                    The Mixed Procedure

                     Model Information

   Data Set                     WORK.TASK
   Dependent Variable           Pulse
   Covariance Structure         Unstructured
   Subject Effect               Replicate
   Estimation Method            REML
   Residual Variance Method     None
   Fixed Effects SE Method      Model-Based
   Degrees of Freedom Method    Between-Within


                  Class Level Information

   Class        Levels    Values

   Group             6    1 2 3 4 5 6
   Replicate        13    1 2 3 4 5 6 7 8 9 10 11 12 13


                       Dimensions

            Covariance Parameters            21
            Columns in X                      7
            Columns in Z                      0
            Subjects                         13
            Max Obs Per Subject               6


                  Number of Observations

          Number of Observations Read              68
          Number of Observations Used              68
          Number of Observations Not Used           0


                     Iteration History

Iteration    Evaluations    -2 Res Log Like       Criterion

        0              1       403.20202450
        1              2       401.06974384      0.00000003
        2              1       401.06974010      0.00000000


                Convergence criteria met.
```

**The Mixed Procedure**

**Estimated R Matrix for Replicate 1**

| Row | Col1 | Col2 | Col3 | Col4 | Col5 | Col6 |
|---|---|---|---|---|---|---|
| 1 | 24.5769 | | | | | |
| 2 | | 32.0833 | | | | |
| 3 | | | 42.0717 | | | |
| 4 | | | | 40.2824 | | |
| 5 | | | | | 21.1050 | |
| 6 | | | | | | 23.9840 |



The Mixed Procedure

**Null Model Likelihood Ratio Test**

| DF | Chi-Square | Pr > ChiSq |
|---|---|---|
| 5 | 2.13 | 0.8306 |

**Type 3 Tests of Fixed Effects**

| Effect | Num DF | Den DF | F Value | Pr > F |
|---|---|---|---|---|
| Group | 5 | 50 | 4.05 | 0.0036 |

Above are the estimated covariance matrix for the six (6) treatment Groups, the test of the covariance matrix differing from the usual assumption of equal variances and the test of the null hypothesis of equal means for the six groups. The estimated variances for each Group are shown on the diagonal of the covariance matrix and range from a low of 21.1050 in Group 5 to a high of 42.0717 in Group 3. The "Null Model Likelihood Ratio Test" is an assessment of the null hypothesis that the variances are all equal versus the alternative hypothesis that the variance structure determined by the "Type = un(1)" statement is a better fit. In this case, the null hypothesis is not rejected (P = 0.8305). The "Type 3 Tests fo Fixed Effects" presents the test of the null hypothesis that the Group means are all equal versus the alternative hypothesis that at least one of the Group means differ, while adjusting for the variances being different.. In this case, the null hypothesis is rejected (P = 0.0035).